

0

A Tutorial Introduction to Maple

Aims and Objectives

- To provide a tutorial guide to Maple.
- To give practical experience in using the package.
- To promote self-help using the online help facilities.
- To provide a concise source of reference for experienced users.

On completion of this chapter, the reader should be able to

- use Maple as a tool;
- produce simple Maple worksheets;
- access some Maple commands and worksheets over the World Wide Web.

It is assumed that the reader is familiar with either the *Windows* or *UNIX* platform. This book was prepared using Maple (Version 13), but most programs should work under earlier and later versions of the package. Note that the online version of the Maple commands for this book will be written using the most up to date version of the package.

The command lines and programs listed in this chapter have been chosen to allow the reader to become familiar with Maple within a few hours. They provide a concise summary of the type of commands that will be used throughout the

text. New users should be able to start on their own problems after completing the chapter, and experienced users should find this chapter an excellent source of reference. Of course, there are many Maple textbooks on the market for those who require further applications or more detail.

If you experience any problems, there are several options for you to take. There is an excellent index within Maple, and Maple commands, worksheets, programs, and output can also be viewed in color over the Web at

<http://www.docm.mmu.ac.uk/STAFF/S.Lynch>

or downloaded at the Maple Application Center

<http://www.maplesoft.com/applications/>.

0.1 A Quick Tour of Maple

To start Maple, simply double-click on the Maple icon. In the Unix environment, one types **maple** as a shell command. The author has used the Windows platform in the preparation of this material. When Maple starts up, a blank worksheet appears on the computer screen entitled **Untitled (1)** and some palettes with buttons appear along the left-hand side. Some examples of palettes are given in Figure 0.1. The buttons on the palettes serve essentially as additional keys on the user's keyboard. Input to the Maple worksheet can either be performed by typing in text commands or pointing and clicking on the symbols provided by various palettes and subpalettes.

Maple has two standard worksheet interfaces: Document mode and Worksheet mode. The Document mode is designed for quick calculations either by typing in commands or using the buttons on the palettes. The user can enter a mathematical expression and then evaluate, manipulate, plot, or solve with a few keystrokes or mouse clicks. Using the Document mode, the user can access Maple without needing to know the Maple syntax. There are two types of content that can be typed in Document mode; these are Text and Math modes. It is important to note that you can only execute a statement if it is entered in Math mode. It is impossible to convey the full functionality of Maple in a book; therefore, it is highly recommended that all users of Maple watch the Maple 13 demo movie at the Maplesoft website (<http://www.maplesoft.com/>). Readers may also be interested in the products Maple Toolbox for MATLAB, Connectivity Toolbox for Simulink, Placement Test Suite, Maple T.A., and MapleNet. The Connectivity Toolbox for Simulink is discussed in Chapter 18.

The Worksheet mode is designed for interactive use through Maple commands, which may offer advanced functionality or customized control not available using context menus or other syntax-free methods. Using either mode, one can create high-quality interactive mathematical documents or presentations.


As this book relies heavily on Maple programming, the author has decided to adopt the Maple Worksheet mode. It is also the mode that the author has used




(a)



(b)



(c)



(d)

Figure 0.1: Some Maple palettes: (a) common symbols; (b) expression; (c) units (FPS); (d) units (SI).

for many years. However, the use of palettes can save some time in typing, and the reader may wish to experiment in the Document mode. To create a Maple document in Worksheet mode, click on **File, New** and choose Worksheet Mode in the Maple window.

Maple can be used to generate full publication-quality documents. In fact, all of the Maple Help pages have been created in either Document mode or Worksheet

mode. The **Help** menu also includes an online version Maplesoft's documentation. The author recommends a brief tour of some of the Help pages to give the reader an idea of how the worksheets can be used. For example, click on the **Help** toolbar at the top of the Maple graphical user interface and scroll down to **Help Maple**. Simply type in **solve** in the Search box and type ENTER; an interactive Maple help page will be opened showing the syntax, some related commands, and examples of the solve command. You can then **Edit** and **Copy** examples into your worksheet.

The author has provided the reader with a tutorial introduction to Maple in Sections 0.2, 0.3, and 0.4. Each tutorial should take no more than one hour to complete. The author highly recommends that new users go through these tutorials line by line; however, readers already familiar with the package will probably use Chapter 0 as reference material only.

Tutorial One provides a basic introduction to the Maple package. The first command line shows the reader how to input comments, which are extremely useful when writing long or complicated programs. The reader will type in `# This is a comment` after the `>` prompt and then type ENTER or RETURN. Maple will label the first input with `> # This is a comment`. Note that no output is given for a comment. The second input line is simple arithmetic. The reader types `1+2-3`; and types ENTER to compute the result. Note that Maple requires a delimiter, either a semicolon to see the output or a colon to suppress the output. Maple labels the second input with `> 1+2-3`; and labels the corresponding output, zero in this case, with `(1)`. As the reader continues to input new command lines, the output numbers change accordingly. This allows users to easily label output that may be useful later in the worksheet. Tutorial Two contains graphic commands and commands used to solve simple differential equations. Tutorial Three provides a simple introduction to programming with Maple.

The tutorials are intended to give the reader a concise and efficient introduction to the Maple package. Many more commands are listed in other chapters of the book, where the output has been included. Of course, there are many Maple textbooks on the market for those who require further applications or more detail. A list of some textbooks is given in the reference section of this chapter.

0.2 Tutorial One: The Basics (One Hour)

There is no need to copy the comments; they are there to help you. Click on the Maple icon and copy the commands. Press ENTER at the end of a line to see the answer. You can interrupt a calculation at any time by clicking on the "Interrupt the current operation" icon in the toolbar. Recall that a working Maple worksheet of Tutorial One can be downloaded from the Maple Application Center at

<http://www.maplesoft.com/applications/>.

Maple Commands**Comments**

> # This is a comment.	# Helps when writing # programs.
> 1+2-3;	# Simple addition and # subtraction.
> 2*3/7;	# Multiplication and # division.
> 2*6+3^2-4/2;	# Powers.
> (5+3)*(4-2);	# Brackets.
> sqrt(100);	# The square root.
> n1:=10:	# The colon suppresses # the output.
> printf("n1=%-d",n1):	# Print the value # of n1.
> n1^(-1);	# Negative powers.
> sin(Pi/3);	# Use capital P for Pi.
> evalf(sin(Pi/3));	# Evaluate as a floating # point number.
> y:=sin(x)+3*x^2;	# Equations and # assignments.
> diff(y,x);	# Differentiate y with # respect to x.
> y:='y':	# Set y back equal to y.
> diff(x^3*y^2,x\$1,y\$2);	# Partial differentiation.
> int(cos(x),x);	# Integration with # respect to x.
> int(x/(x^3-1),x=0..1);	# Definite integration.
> int(1/x,x=1..infinity);	# Improper integration.
> convert(1/((s+1)*(s+2)),parfrac,s);	# Split into partial # fractions.
> expand(sin(x+y));	# Expansion.
> factor(x^2-y^2);	# Factorization.
> limit((cos(x)-1)/x,x=0);	# The limit as x goes # to zero.


```

> invlaplace(6/s^4,s,t);           # Inverse transform.
> fourier(t^4*exp(-t^2),t,w);      # Fourier transform.
> invfourier(%,w,t);               # Transform previous line.
> ??coeff                           # Open a help page for
                                   # coeff.
> ???coeff                          # List the syntax for
                                   # this command.
> ???coeff                          # List some examples.
> # End of Tutorial One.

```

0.3 Tutorial Two: Plots and Differential Equations (One Hour)

Maple has excellent graphical capabilities and many solutions of nonlinear systems are best portrayed graphically. The graphs produced from the input text commands listed below may be found in the Tutorial Two worksheet, which can be downloaded from the Maple Application Center. Plots in other chapters of the book are referred to in many of the Maple programs at the end of each chapter.

```

> # Plotting graphs.

> # Set up the domain and plot a simple function.
> plot(cos(2*x),x=0..4*Pi,font=[TIMES,ROMAN,20],color=black);

> # Plot two curves on one graph.
> plot({x*cos(x),x-2},x=-5..5);

> # Plotting with a title.
> plot(x^3,x=-3..3,y=-30..30,title='A cubic polynomial');

> # Plotting with discontinuities.
> plot(tan(x),x=-2*Pi..2*Pi,y=-10..10,discont=true);

> # Plotting with different line styles.
> c1:=plot(sin(x),x=-2*Pi..2*Pi,linestyle=1):
> c2:=plot(2*sin(2*x-Pi/2),x=-2*Pi..2*Pi,linestyle=3):
> display({c1,c2});

> # Plotting points.
> points:=[n,sin(n)]$n=1..10]:
> plot(points,x=0..15,style=point,symbol=circle);

```

```

> # Surface plot.
> plot3d(y^2*x^2/2+x^2*y^2/2-x^2/2-y^2/2,x=-2..2,y=-2..2,
  axes=boxed);

> # A contour plot.
> contourplot(y^2*x^2/2+x^2*y^2/2-x^2/2-y^2/2,x=-2..2,y=-2..2,
  contours=50,grid=[50,50]);

> # A cylinder plot.
> cylinderplot(z+3*cos(2*theta),theta=0..Pi,z=0..3);

> # An implicit plot.
> implicitplot(y^2+y=x^3-x,x=-2..3,y=-3..3,numpoints=1000);

> # Solving simple ODEs analytically and numerically.
> # Load the differential equations package.
> with(DEtools):

> # Solve a simple ODE.
> dsolve(diff(y(x),x)=x,y(x));

> # Solve an initial value problem (IVP).
> dsolve({diff(v(t),t)+2*t=0,v(1)=5},v(t));

> # Solve a second order ODE.
> dsolve(diff(x(t),t$2)+8*diff(x(t),t)+25*x(t)=0,x(t));

> # Plot a solution curve for a second order ODE.
> deqn:=diff(y(x),x$2)=x^3*y(x)+1;
> DEplot(deqn,y(x),x=-5..2,[[y(0)=0.5,D(y)(0)=1]],linecolor=blue,
  thickness=1);

> # Plot the solution curve for a stiff van der Pol system of ODEs.
> mu:=1000:
> deq:=diff(y(x),x,x)-mu*(1-y(x)^2)*diff(y(x),x)+y(x)=0:
> ics:={y(0)=2,D(y)(0)=0}:
> dsol:=dsolve({deq} union ics,numeric,range=0..3000,stiff=true ):
> plots[odeplot](dsol,[x,y(x)]);

```

0.4 Simple Maple Programs

Sections 0.1, 0.2, and 0.3 illustrate the interactive nature of Maple. More involved tasks will require more code. Each Maple program is displayed between horizontal lines and kept short to aid in understanding; the output is also included. Type SHIFT-ENTER at the end of a command line so that the program will execute on one ENTER command.

In Document mode, the reader should use the Exploration assistant for Interactive Exploration.

Procedures. Declare local variables within procedures.

```
> # Program 1: Procedures.
> # The norm of a 3-dimensional vector.
> norm3d:=proc() local a,b,c;sqrt(a^2+b^2+c^2) end;
> norm3d(3,4,5);
```

$5\sqrt{2}$

```
> # Program 2: For..do..end loop.
> # Sum the natural numbers from 1 to 100.
> i:='i':total:=0:
  for i from 0 to 100 do
    total:=i+total:
  end do:
  total;
```

5050

```
> # Program 3: If..then..elif...else.
> # Determine if p is less than or not less than 2.
> p:=4:
  if p<2 then printf("p is less than 2");
  elif p>=2 then printf("p is not less than 2");
  end if;
```

p is not less than 2

```
> # Program 4: Arrays and sequences.
> # List the first 10 terms of the Fibonacci sequence.
> F:=array(1..10000):
  F[1]:=1:F[2]:=1:N:=10:
  for i from 3 to N do
    F[i]:=F[i-1]+F[i-2]:
  end do:
  seq(F[i],i=1..N);
```

{1, 1, 2, 3, 5, 8, 13, 21, 34, 55}

```
> # Program 5: Iteration.
> # List the last 10 iterates of the logistic map.
> mu:=3.2:x[0]:=0.2:
  for n from 0 to 99 do
```

```

x[n+1]:=mu*x[n]*(1-x[n]):
end do:
for n from 90 to 99 do
nprintf("x[%-d]=%g",n+1,x[n+1]);
end do;

x[91]=0.513044, x[92]=0.799455, x[93]=0.513044, x[94]=0.799455,
x[95]=0.513044, x[96]=0.799455, x[97]=0.513044, x[98]=0.799455,
x[99]=0.513044, x[100]=0.799455

```

```

> # Program 6: Multiple plots with text.
> # Figure 0.2: Plot solution curves to ODEs.
> # Note that it is sometimes better to label colored curves with colored
> # text, as below. Figure 0.2 in the book is a black and white version.
> deqn1:=diff(x(t),t$2)=-2*diff(x(t),t)-25*x(t):
p3:=DEplot(deqn1,x(t),t=0..10,[[x(0)=1,D(x)(0)=0]],stepsize=0.1,
linecolor=blue,thickness=1):
deqn2:=diff(x(t),t$2)=-25*x(t):
p4:=DEplot(deqn2,x(t),t=0..10,[[x(0)=1,D(x)(0)=0]],stepsize=0.1,
linecolor=red,thickness=1):
t3:=textplot([6,1,'Harmonic motion'],align=RIGHT,color=red):
t4:=textplot([1.8,0.2,'Damped motion'],align=RIGHT,color=blue):
display({p3,p4,t3,t4},labels=['t','x']);

```




Figure 0.2: Harmonic and damped motion of a pendulum.

```

> # Program 7: Interactive exploration. Intersecting curves.
> # On execution of the command, an interactive parameter Maplet
> # pops up with a parameter slider. The plot varies as the slider is
> # moved up and down.
> restart:with(DEtools):with(plots):
interactiveparams(plot,[{1-a*x,(x-1)/a},x=0..2],a=0..1);

```

```

> # Program 8: Interactive exploration. Solutions to a system of
> # differential equations.
> # On execution of the command, an interactive parameter Maplet
> # pops up with two parameter sliders. The plot varies as the sliders
> # are moved up and down.
> restart:with(DEtools):with(plots):
  sys1:=diff(x(t),t)=-a*x(t),diff(y(t),t)=a*x(t)-b*y(t),diff(z(t),t)
    =b*y(t);
  dsolve([sys1,x(0)=10,y(0)=0,z(0)=0]);
  interactiveparams(plot,[{10*exp(-a*t),
    10*a*exp(-b*t)/(a-b)-10*a*exp(-a*t)/(a-b),
    (10*exp(-a*t)*b-10*exp(-b*t)*a^2/(a-b)+
    10*exp(-b*t)*a*b/(a-b)+10*a-10*b)/(a-b)},t=0..30],
    a=0.1..0.9,b=0.1..0.8);

```




Figure 0.3: Solution curves for Program 8 when $a = 0.5$ and $b = 0.45$.

0.5 Hints for Programming

The Maple language contains very powerful commands, which means that some complex programs may contain only a few lines of code. Of course, the only way to learn programming is to sit down and try it yourself. This section has been included to point out common errors and give advice on how to troubleshoot. Remember

to check the Help pages in Maple and the Web if the following does not help you with your particular problem.

Common typing errors. The author strongly advises new users to type Tutorials One, Two, and Three into their own worksheets; this should reduce typing errors.

- All command lines must end with a colon or semicolon.
- Type ENTER at the end of every command line.
- If a command line is ended with a colon, the output will not be displayed.
- Make sure brackets, parentheses, etc. match up in correct pairs.
- Remember Maple is case sensitive.
- Check the syntax; type `??solve` to list syntax for the solve command, for example.

Programming tips. The reader should use the Maple programs listed in Section 0.4 to practice simple programming techniques.

- Use the `restart` command at the beginning of a new program.
- Use comments throughout a program. You will find them extremely useful in the future.
- Use procedures to localize variables. This is especially useful for very large programs.
- If a program involves a large number of iterations (e.g., 50,000), then run it for three iterations first and list all output.
- If the computer is not responding, click on the interrupt icon and try reducing the size of the problem.
- Read the error message printed by Maple.
- Find a similar Maple program in a book or on the Web and edit it to meet your needs.
- Check which version of Maple you are using. The syntax of some commands may have altered. For example, some Maple 11 programs will not run under Maple 13.

0.6 Maple Exercises

1. Evaluate the following:

- (a) $4 + 5 - 6$;
- (b) 3^{12} ;
- (c) $\sin(0.1\pi)$;
- (d) $(2 - (3 - 4(3 + 7(1 - (2(3 - 5))))))$;
- (e) $\frac{2}{5} - \frac{3}{4} \times \frac{2}{3}$.

2. Given that

$$A = \begin{pmatrix} 1 & 2 & -1 \\ 0 & 1 & 0 \\ 3 & -1 & 2 \end{pmatrix}, \quad B = \begin{pmatrix} 1 & 2 & 3 \\ 1 & 1 & 2 \\ 0 & 1 & 2 \end{pmatrix}, \quad C = \begin{pmatrix} 2 & 1 & 1 \\ 0 & 1 & -1 \\ 4 & 2 & 2 \end{pmatrix},$$

determine the following:

- (a) $A + 4BC$;
- (b) the inverse of each matrix if it exists;
- (c) A^3 ;
- (d) the determinant of C ;
- (e) the eigenvalues and eigenvectors of B .

3. Given that $z_1 = 1 + i$, $z_2 = -2 + i$, and $z_3 = -i$, evaluate the following:

- (a) $z_1 + z_2 - z_3$;
- (b) $\frac{z_1 z_2}{z_3}$;
- (c) e^{z_1} ;
- (d) $\ln(z_1)$;
- (e) $\sin(z_3)$.

4. Evaluate the following limits if they exist:

- (a) $\lim_{x \rightarrow 0} \frac{\sin x}{x}$;
- (b) $\lim_{x \rightarrow \infty} \frac{x^3 + 3x^2 - 5}{2x^3 - 7x}$;
- (c) $\lim_{x \rightarrow \pi} \frac{\cos x + 1}{x - \pi}$;
- (d) $\lim_{x \rightarrow 0^+} \frac{1}{x}$;
- (e) $\lim_{x \rightarrow 0} \frac{2 \sinh x - 2 \sin x}{\cosh x - 1}$.

5. Find the derivatives of the following functions:

(a) $y = 3x^3 + 2x^2 - 5$;

(b) $y = \sqrt{1 + x^4}$;

(c) $y = e^x \sin x \cos x$;

(d) $y = \tanh x$;

(e) $y = x^{\ln x}$.

6. Evaluate the following definite integrals:

(a) $\int_{x=0}^1 3x^3 + 2x^2 - 5 \, dx$;

(b) $\int_{x=1}^{\infty} \frac{1}{x^2} \, dx$;

(c) $\int_{-\infty}^{\infty} e^{-x^2} \, dx$;

(d) $\int_0^1 \frac{1}{\sqrt{x}} \, dx$;

(e) $\int_0^{\frac{2}{\pi}} \frac{\sin(1/t)}{t^2} \, dt$.

7. Graph the following:

(a) $y = 3x^3 + 2x^2 - 5$;

(b) $y = e^{-x^2}$, for $-5 \leq x \leq 5$;

(c) $x^2 - 2xy - y^2 = 1$;

(d) $z = 4x^2e^y - 2x^4 - e^{4y}$, for $-3 \leq x \leq 3$ and $-1 \leq y \leq 1$;

(e) $x = t^2 - 3t$, $y = t^3 - 9t$, for $-4 \leq t \leq 4$.

8. Solve the following differential equations:

(a) $\frac{dy}{dx} = \frac{x}{2y}$, given that $y(1) = 1$;

(b) $\frac{dy}{dx} = \frac{-y}{x}$, given that $y(2) = 3$;

(c) $\frac{dy}{dx} = \frac{x^2}{y^3}$, given that $y(0) = 1$;

(d) $\frac{d^2x}{dt^2} + 5\frac{dx}{dt} + 6x = 0$, given that $x(0) = 1$ and $\dot{x}(0) = 0$;

(e) $\frac{d^2x}{dt^2} + 5\frac{dx}{dt} + 6x = \sin(t)$, given that $x(0) = 1$ and $\dot{x}(0) = 0$.

9. Carry out 100 iterations on the recurrence relation

$$x_{n+1} = 4x_n(1 - x_n),$$

given that (a) $x_0 = 0.2$ and (b) $x_0 = 0.2001$. List the final 10 iterates in each case.

10. Type `?while` to read the help page on the while command. Use a while loop to program Euclid's algorithm for finding the greatest common divisor of two integers. Use your program to find the greatest common divisor of 12,348 and 14,238.

Recommended Textbooks

Note that Maple documentation comes with the package and is also available through the Help pages. More Maple books are listed in the reference sections of other chapters in the book.

- [1] D. Richards, *Advanced Mathematical Methods with Maple*, 2nd ed., Cambridge University Press, Cambridge, 2009.
- [2] B. Barnes and G. R. Fulford, *Mathematical Modelling with Case Studies: A Differential Equations Approach using Maple and MATLAB*, 2nd ed., Chapman and Hall, London, 2008.
- [3] I. K. Shingareva and C. Lizárraga-Celaya, *Maple and Mathematica: A Problem Solving Approach for Mathematics*, Springer-Verlag, New York, 2007.
- [4] M. L. Abell and J. P. Braselton, *Maple By Example*, 3rd ed., Academic Press, New York, 2005.
- [5] A. Heck, *Introduction to Maple*, 3rd ed., Springer-Verlag, New York, 2003.